QUERY 1

“Return the average Squat Bench Deadlift for raw and equipped categories”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>

SELECT   (AVG(?rawSquat) AS ?avgRawSquat) ?avgEqSquat (AVG(?rawBench) AS ?avgRawBench)
?avgEqBench (AVG(?rawDeadlift) AS ?avgRawDeadlift) ?avgEqDeadlift

WHERE

{

   VALUES ?raw {:Wraps :Raw}

   VALUES ?eq {:Straps :Single-ply :Multi-ply}


   ?athleteRaw :did ?competitionRaw.

   ?competitionRaw :bestSquat ?rawSquat;

              :bestBench ?rawBench;

              :bestDeadlift ?rawDeadlift;

              :equipment ?raw.

   {

      SELECT (AVG(?eqSquat) AS ?avgEqSquat) (AVG(?eqBench) AS ?avgEqBench)

            (AVG(?eqDeadlift) AS ?avgEqDeadlift) where {


           ?athleteEq :did ?competitionEq.

           ?competitionEq :bestSquat ?eqSquat;

                      :bestBench ?eqBench;

                      :bestDeadlift ?eqDeadlift;

                      :equipment ?eq.

      }

}

}

GROUP BY ?avgEqSquat ?avgEqBench ?avgEqDeadlift
```

QUERY 2

“Return for each country the average squat bench deadlift, having a number of competition for that country greater than 30 ordered by descending”

```
PREFIX pl: <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>


SELECT ?countryName (AVG(?squat) AS ?avgSquat) (AVG(?bench) AS ?avgBench)
                        (AVG(?deadlift) AS ?avgDeadlift) (AVG(?totalKg) AS ?avgTotalKg) (COUNT(?athlete)
AS ?totAthlete) WHERE

{

        ?athlete pl:nationality ?country;

                pl:did ?competition.

        ?country pl:name ?countryName.

        ?competition pl:IPFClass ?category;

                        pl:bestSquat ?squat;

                        pl:bestBench ?bench;

                        pl:bestDeadlift ?deadlift;

                        pl:totalKg ?totalKg;

}

GROUP BY (?countryName)

HAVING(?totAthlete > 30)

ORDER BY DESC (?avgTotalKg)

limit 10
```

QUERY 3

“Return the heaviest attempt (even failed) for squat bench deadlift at a competition”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>

select (MAX(?absSquat) AS ?maxSquatAttempt) (MAX(?absBench) AS ?maxBenchAttempt)
                                                    (MAX(?absDeadlift) AS ?maxDeadliftAttempt) where {

    {

        SELECT (ABS(?squat) AS ?absSquat) (ABS(?bench) AS ?absBench) (ABS(?deadlift) AS ?absDeadlift)
        WHERE {

                ?athlete :did ?competition.

                ?competition :1stSquat | :2ndSquat | :3rdSquat | :4thSquat | :bestSquat ?squat;

                                :1stBench | :2ndBench | :3rdBench | :4thBench | :bestBench ?bench;

                            :1stDeadlift | :2ndDeadlift | :3rdDeadlift | :4thDeadlift | :bestDeadlift ?deadlift.

         }

    }

} LIMIT 10
```

QUERY 4

“Return for each athlete their personal best for squat bench deadlift and total Kg ordered by descending totKg”

```
PREFIX pl: <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>

PREFIX foaf: <http://xmlns.com/foaf/0.1/>

SELECT  ?athleteName (MAX(?squat) AS ?bestSquat) (MAX(?bench) AS ?bestBench)

                (MAX(?deadlift) AS ?bestDeadlift) (MAX(?totalKg) AS ?bestTotKg)  WHERE

{

        ?athlete pl:did ?competition;

                foaf:name ?athleteName.

        ?competition pl:totalKg ?totalKg;

                pl:bestSquat ?squat;

                pl:bestBench ?bench;

                pl:bestDeadlift ?deadlift;

}

GROUP BY (?athleteName)

ORDER BY DESC (?bestTotKg)

LIMIT 10
```

QUERY 5

“Return the name , age, weight and scores of the athlete with higher IPFScore or wilksScore or mcCullochScore or glossbrennerScore”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>

PREFIX foaf: <http://xmlns.com/foaf/0.1/>


SELECT ?name ?age ?weight ?IPFScore ?wilksScore ?mcCullochScore ?glossbrennerScore WHERE

{

   ?athlete :did ?competition;

                foaf:name ?name.

   ?competition :IPFScore ?IPFScore;

          :wilksScore ?wilksScore;

          :mcCullochScore ?mcCullochScore;

          :glossbrennerScore ?glossbrennerScore.

   OPTIONAL{?competition :athleteAge ?age;}

   OPTIONAL{?competition :athleteWeight ?weight.}

   FILTER(?IPFScore >= ?maxIPFScore || ?wilksScore >= ?maxWilksScore || ?mcCullochScore >=
                ?maxCullochScore || ?glossbrennerScore >= ?maxGlossbrennerScore)

         {

                SELECT (MAX(?IPFScore) AS ?maxIPFScore) (MAX(?wilksScore) AS ?maxWilksScore)

                (MAX(?mcCullochScore) AS ?maxCullochScore) (MAX(?glossbrennerScore) AS
                ?maxGlossbrennerScore)

                WHERE {

                        ?competition :IPFScore ?IPFScore;

                                        :wilksScore ?wilksScore;

                                        :mcCullochScore ?mcCullochScore;

                                        :glossbrennerScore ?glossbrennerScore.

                }

         }

}

LIMIT 10
```

QUERY 6

“For each age class group by sex and return the average total kg”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>
SELECT ?athleteSex ?ageClass (AVG(?totKg) AS ?avgTotKg) WHERE {
        ?athlete :did ?competition;
                :sex ?athleteSex.
        ?competition :IPFAgeClass ?ageClass;
                        :totalKg ?totKg.
}
GROUP BY  ?athleteSex ?ageClass
ORDER BY DESC(?avgTotKg)
LIMIT 100
```

QUERY 7

“Return the average number of athlete at a meeting for each country”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>
SELECT ?countryName (AVG(?totAthlete) AS ?avgTotAthlete) WHERE {
        ?meet :nationality ?country.
        ?country :name ?countryName
{
        SELECT ?meet (COUNT(?ath) AS ?totAthlete) WHERE{
                ?competition :doneDuring ?meet.
                ?ath :did ?competition.
        }
   GROUP BY (?meet)
}
}
GROUP BY (?countryName)
ORDER BY DESC (?avgTotAthlete)
LIMIT 100
```

QUERY 8

“Return the average number of meetings attended by an athlete in a single year”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>
SELECT (AVG(?totalMeet) AS ?avgTotMeet) WHERE
{
    SELECT ?athlete ?year (COUNT(?competition) AS ?totalMeet) WHERE {
       ?athlete :did ?competition.
       ?competition :doneDuring ?meet.
       ?meet :year ?year.
    }
    GROUP BY ?athlete ?year
    ORDER BY DESC (?totalMeet)
}

LIMIT 100
```

QUERY 9

“For each type of event, compare the average lift between the competitions done in the same country of the athlete against the ones done in a country different from the athlete”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>

SELECT ?event (AVG(?hTotKg) AS ?homeAvg) (AVG(?aTotKg) AS ?abroadAvg)

WHERE {
        ?ath :did ?hComp;
                :did ?aComp;
                :nationality ?athCountry.
        ?hComp :totalKg ?hTotKg;
                :inEvent ?event;
                :doneDuring ?hMeeting.
        ?hMeeting :nationality ?hmCountry.
        FILTER(?athCountry = ?hmCountry)
        ?aComp :totalKg ?aTotKg;
                :inEvent ?event;
                :doneDuring ?aMeeting.
        ?aMeeting :nationality ?amCountry.
        FILTER(?athCountry != ?amCountry)
}
GROUP BY ?event
limit 20
```

QUERY 10

“Return the ten most recurrent encounters of a couple of athletes during their various competitions”

```
PREFIX : <http://www.semanticweb.org/mario/ontologies/2022/10/powerlifting#>
PREFIX cns: <http://eulersharp.sourceforge.net/2003/03swap/countries#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT ?name1 ?name2 (COUNT(?comp1) AS ?Nduels)
WHERE {
        ?ath1 :did ?comp1;
                foaf:name ?name1.
        ?comp1 :doneDuring ?meet;
                :inEvent ?event;
                :IPFClass ?cat.
        ?ath2 :did ?comp2;
                foaf:name ?name2.
        ?comp2 :doneDuring ?meet;
                :inEvent ?event;
                :IPFClass ?cat.
        FILTER(?ath1 != ?ath2 )
}
GROUP BY ?name1 ?name2
ORDER BY DESC(?Nduels)
limit 10
```